Numerical Method for 3D Ideal Flow

Author: Han de Bruijn
Dated: 2010 April

The two dimensional analogue of an octahedron is a parallelogram. However, the obvious generalization of a parallelogram, being a quadrilateral, seems to be a hexahedron. This explains why it has lasted so long - six years - before I found a generalization of 2D ideal ow to three dimensions. A Least Squares Finite Element Method for two dimensional incompressible and irrotational (i.e ideal) ow has been described, as "Labrujere's Problem", at:

http://hdebruijn.soo.dto.tudelft.nl/jaar2004/nlrlsfem.pdf

This document is one of the absolute prerequisites for the 3D study below.
Summary. We start with the algebraic description of a parent hexahedron (cube). This well known Finite Element is to be employed later on as a exible building block (brick) in 3D meshes. Next we describe the well known Finite Di erence seven node star = parent F.E. octahedron. The equations for Ideal Flow are discretized at octahedrons inside hexahedrons. Source code and results of a Patch Test are included.

Parent Hexahedron



The hexahedron is a Finite Element which is de ned in its parent (i.e. normed) coordinates (ξ, η, ζ) as a shape with eight nodes:

$$\begin{aligned}
(0) &= (-1,-1,-1) = (\xi_0,\eta_0,\zeta_0) \\
(1) &= (+1,-1,-1) = (\xi_1,\eta_1,\zeta_1) \\
(2) &= (-1,+1,-1) = (\xi_2,\eta_2,\zeta_2) \\
(3) &= (+1,+1,-1) = (\xi_3,\eta_3,\zeta_3) \\
(4) &= (-1,-1,+1) = (\xi_4,\eta_4,\zeta_4) \\
(5) &= (+1,-1,+1) = (\xi_5,\eta_5,\zeta_5) \\
(6) &= (-1,+1,+1) = (\xi_6,\eta_6,\zeta_6) \\
(7) &= (+1,+1,+1) = (\xi_7,\eta_7,\zeta_7)
\end{aligned}$$

Template for above picture copied without permission from:

`http://www.metafysica.nl/turing/hexaedra_regularia_a.gif`

Why this numbering? Answer: because it's systematical. And therefore it can be generalized easily to multiple dimensions. Just replace -1 by 0 and leave $+1 = 1$ unchanged everywhere. Then reverse the bits. Et voila: what we have is the binary representation of our numbering $0 \ldots 7$. When formulated in a well known programming language, it sounds as follows:

```
function nr(i,j,k : integer) : integer; { F.E. node numbering }
begin
  nr := ((k+1) div 2)*4 + ((j+1) div 2)*2 + ((i+1) div 2) * 1;
end;
```

With a Finite Element Method, eight so-called shape functions $N_m(\xi, \eta, \zeta)$ should be de ned, in such way that:

$$\begin{cases} N_m(\xi_k,\eta_k,\zeta_k) = 1 & \text{for} \quad k = m \\ N_m(\xi_k,\eta_k,\zeta_k) = 0 & \text{for} \quad k \neq m \end{cases}$$

Such nite element shape functions have been found for the one-dimensional and two-dimensional analogues of the hexahedron: line segment and quadrilateral. The following 2D study shall be considered as a prerequisite for the 3D case:

`http://hdebruijn.soo.dto.tudelft.nl/jaar2004/vierhoek.pdf`

After absorbing this reference, it shouldn't be di cult to make an educated guess for the shape functions of the hexahedron:

$$\begin{aligned}
N_0(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1-\zeta) \\
N_1(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1-\zeta) \\
N_2(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1-\zeta)
\end{aligned}$$

$$
\begin{aligned}
N_3(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1-\zeta)\\
N_4(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1+\zeta)\\
N_5(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1+\zeta)\\
N_6(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1+\zeta)\\
N_7(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1+\zeta)
\end{aligned}
$$

As employed in:

$$f = N_0 f_0 + N_1 f_1 + N_2 f_2 + N_3 f_3 + N_4 f_4 + N_5 f_5 + N_6 f_6 + N_7 f_7$$

Instead we can collect terms belonging to $(1,\xi,\eta,\zeta,\xi\eta,\xi\zeta,\eta\zeta,\xi\eta\zeta)$. It involves a bit of work, but then you have some:

$$
\begin{aligned}
f(\xi,\eta,\zeta) &= \frac{1}{8}(+f_0+f_1+f_2+f_3+f_4+f_5+f_6+f_7)\\
&+ \frac{1}{8}(-f_0+f_1-f_2+f_3-f_4+f_5-f_6+f_7)\,\xi\\
&+ \frac{1}{8}(-f_0-f_1+f_2+f_3-f_4-f_5+f_6+f_7)\,\eta\\
&+ \frac{1}{8}(-f_0-f_1-f_2-f_3+f_4+f_5+f_6+f_7)\,\zeta\\
&+ \frac{1}{8}(+f_0-f_1-f_2+f_3+f_4-f_5-f_6+f_7)\,\xi\eta\\
&+ \frac{1}{8}(+f_0-f_1+f_2-f_3-f_4+f_5-f_6+f_7)\,\xi\zeta\\
&+ \frac{1}{8}(+f_0+f_1-f_2-f_3-f_4-f_5+f_6+f_7)\,\eta\zeta\\
&+ \frac{1}{8}(-f_0+f_1+f_2-f_3+f_4-f_5-f_6+f_7)\,\xi\eta\zeta
\end{aligned}
$$

The truncated Taylor Series expansion of $f(\xi,\eta,\zeta)$ is:

$$
\begin{aligned}
f(\xi,\eta,\zeta) &= f(0,0,0)\\
&+ \frac{\partial f}{\partial\xi}(0,0,0)\,\xi + \frac{\partial f}{\partial\eta}(0,0,0)\,\eta + \frac{\partial f}{\partial\zeta}(0,0,0)\,\zeta\\
&+ \frac{\partial^2 f}{\partial\xi\partial\eta}(0,0,0)\,\xi\eta + \frac{\partial^2 f}{\partial\xi\partial\zeta}(0,0,0)\,\xi\zeta + \frac{\partial^2 f}{\partial\eta\partial\zeta}(0,0,0)\,\eta\zeta\\
&+ \frac{\partial^3 f}{\partial\xi\partial\eta\partial\zeta}(0,0,0)\,\xi\eta\zeta
\end{aligned}
$$

From which we conclude that (normed, central) Finite Di erence Schemes for the hexahedron are given by:

$$f(0,0,0) = \frac{1}{8}(+f_0+f_1+f_2+f_3+f_4+f_5+f_6+f_7)$$

$$
\begin{aligned}
\frac{\partial f}{\partial \xi}(0,0,0) &= \frac{1}{8}(-f_0 + f_1 - f_2 + f_3 - f_4 + f_5 - f_6 + f_7) \\
\frac{\partial f}{\partial \eta}(0,0,0) &= \frac{1}{8}(-f_0 - f_1 + f_2 + f_3 - f_4 - f_5 + f_6 + f_7) \\
\frac{\partial f}{\partial \zeta}(0,0,0) &= \frac{1}{8}(-f_0 - f_1 - f_2 - f_3 + f_4 + f_5 + f_6 + f_7) \\
\frac{\partial^2 f}{\partial \xi \partial \eta}(0,0,0) &= \frac{1}{8}(+f_0 - f_1 - f_2 + f_3 + f_4 - f_5 - f_6 + f_7) \\
\frac{\partial^2 f}{\partial \xi \partial \zeta}(0,0,0) &= \frac{1}{8}(+f_0 - f_1 + f_2 - f_3 - f_4 + f_5 - f_6 + f_7) \\
\frac{\partial^2 f}{\partial \eta \partial \zeta}(0,0,0) &= \frac{1}{8}(+f_0 + f_1 - f_2 - f_3 - f_4 - f_5 + f_6 + f_7) \\
\frac{\partial^3 f}{\partial \xi \partial \eta \partial \zeta}(0,0,0) &= \frac{1}{8}(-f_0 + f_1 + f_2 - f_3 + f_4 - f_5 - f_6 + f_7)
\end{aligned}
$$

The above can be written in matrix form, as follows. Let:

$$f(\xi,\eta,\zeta) = a_0 + a_1\,\xi + a_2\,\eta + a_3\,\zeta + a_4\,\xi\eta + a_5\,\xi\zeta + a_6\,\eta\zeta + a_7\,\xi\eta\zeta$$

Then a_k are the (normed, central) Finite Di erence schemes:

$$
\begin{bmatrix} a_0 \\ a_1 \\ a_2 \\ a_3 \\ a_4 \\ a_5 \\ a_6 \\ a_7 \end{bmatrix}
= \frac{1}{8}
\begin{bmatrix}
+1 & +1 & +1 & +1 & +1 & +1 & +1 & +1 \\
-1 & +1 & -1 & +1 & -1 & +1 & -1 & +1 \\
-1 & -1 & +1 & +1 & -1 & -1 & +1 & +1 \\
-1 & -1 & -1 & -1 & +1 & +1 & +1 & +1 \\
+1 & -1 & -1 & +1 & +1 & -1 & -1 & +1 \\
+1 & -1 & +1 & -1 & -1 & +1 & -1 & +1 \\
+1 & +1 & -1 & -1 & -1 & -1 & +1 & +1 \\
-1 & +1 & +1 & -1 & +1 & -1 & -1 & +1
\end{bmatrix}
\begin{bmatrix} f_0 \\ f_1 \\ f_2 \\ f_3 \\ f_4 \\ f_5 \\ f_6 \\ f_7 \end{bmatrix}
$$

It is noted that all of the columns are mutually orthogonal. The inverse of an orthogonal matrix is the transpose of the same matrix, apart from a constant. This constant is the inverse of the length of (one of) the column vectors, which in our case is 8. Consequently:

$$
\begin{bmatrix} f_0 \\ f_1 \\ f_2 \\ f_3 \\ f_4 \\ f_5 \\ f_6 \\ f_7 \end{bmatrix}
=
\begin{bmatrix}
+1 & -1 & -1 & -1 & +1 & +1 & +1 & -1 \\
+1 & +1 & -1 & -1 & -1 & -1 & +1 & +1 \\
+1 & -1 & +1 & -1 & -1 & +1 & -1 & +1 \\
+1 & +1 & +1 & -1 & +1 & -1 & -1 & -1 \\
+1 & -1 & -1 & +1 & +1 & -1 & -1 & +1 \\
+1 & +1 & -1 & +1 & -1 & +1 & -1 & -1 \\
+1 & -1 & +1 & +1 & -1 & -1 & +1 & -1 \\
+1 & +1 & +1 & +1 & +1 & +1 & +1 & +1
\end{bmatrix}
\begin{bmatrix} a_0 \\ a_1 \\ a_2 \\ a_3 \\ a_4 \\ a_5 \\ a_6 \\ a_7 \end{bmatrix}
$$

So the nodal values of the nite element shape functions can be expressed in the normed, central nite di erence schemes and also the other way around.

Parent Octahedron





The octahedron is a Finite Difference molecule which is defined in its parent (i.e. normed) coordinates (ξ, η, ζ) as a F.D. star with seven nodes:

$$
\begin{array}{lll}
(0) = (0,0,0) & (1) = (-1,0,0) & (2) = (+1,0,0) \\
 & (3) = (0,-1,0) & (4) = (0,+1,0) \\
 & (5) = (0,0,-1) & (6) = (0,0,+1)
\end{array}
$$

Templates for above pictures copied without permission from:

`http://en.wikipedia.org/wiki/File:Octahedron.svg`
`http://en.wikipedia.org/wiki/File:Octahedral-3D-balls.png`

The (central, normed) Finite Difference schemes associated with the seven node star are well known:

$$
\begin{array}{lll}
f(0,0,0) = f_0 & \dfrac{\partial f}{\partial \xi}(0,0,0) = \dfrac{f_2 - f_1}{2} & \dfrac{\partial^2 f}{\partial \xi^2}(0,0,0) = f_2 - 2f_0 + f_1 \\
 & \dfrac{\partial f}{\partial \eta}(0,0,0) = \dfrac{f_4 - f_3}{2} & \dfrac{\partial^2 f}{\partial \eta^2}(0,0,0) = f_4 - 2f_0 + f_3 \\
 & \dfrac{\partial f}{\partial \zeta}(0,0,0) = \dfrac{f_6 - f_5}{2} & \dfrac{\partial^2 f}{\partial \zeta^2}(0,0,0) = f_6 - 2f_0 + f_5
\end{array}
$$

The Finite Difference interpolation of a function $f(\xi, \eta, \zeta)$ at the molecule is given by the first few terms of a Taylor series expansion:

$$
\begin{aligned}
f(\xi,\eta,\zeta) = f(0,0,0) \quad &+ \quad \frac{\partial f}{\partial \xi}(0,0,0)\xi \; + \; \frac{\partial^2 f}{\partial \xi^2}(0,0,0)\frac{1}{2}\xi^2 \\
&+ \quad \frac{\partial f}{\partial \eta}(0,0,0)\eta \; + \; \frac{\partial^2 f}{\partial \eta^2}(0,0,0)\frac{1}{2}\eta^2 \\
&+ \quad \frac{\partial f}{\partial \zeta}(0,0,0)\zeta \; + \; \frac{\partial^2 f}{\partial \zeta^2}(0,0,0)\frac{1}{2}\zeta^2
\end{aligned}
$$

Herefrom we easily nd (F.D. representation):

$$\begin{aligned} f(\xi,\eta,\zeta) = f_0 \quad &+ \quad \frac{f_2 - f_1}{2}\,\xi \;+\; (f_2 - 2f_0 + f_1)\,\frac{1}{2}\xi^2 \\ &+ \quad \frac{f_4 - f_3}{2}\,\eta \;+\; (f_4 - 2f_0 + f_3)\,\frac{1}{2}\eta^2 \\ &+ \quad \frac{f_6 - f_5}{2}\,\zeta \;+\; (f_6 - 2f_0 + f_5)\,\frac{1}{2}\zeta^2 \end{aligned}$$

It is much less well known that *Finite Element shape functions* hence emerge as coe cients of the function values at the nodes. This is derived from the F.D. representation by collecting terms in a slightly di erent way:

$$\begin{aligned} f(\xi,\eta,\zeta) &= (1 - \xi^2 - \eta^2 - \zeta^2)\, f_0 \\ &+ (-\frac{1}{2}\xi + \frac{1}{2}\xi^2)\, f_1 \;+\; (+\frac{1}{2}\xi + \frac{1}{2}\xi^2)\, f_2 \\ &+ (-\frac{1}{2}\eta + \frac{1}{2}\eta^2)\, f_3 \;+\; (+\frac{1}{2}\eta + \frac{1}{2}\eta^2)\, f_4 \\ &+ (-\frac{1}{2}\zeta + \frac{1}{2}\zeta^2)\, f_5 \;+\; (+\frac{1}{2}\zeta + \frac{1}{2}\zeta^2)\, f_6 \end{aligned}$$

Therefore the F.E. shape functions of the seven node molecule are found to be:

$$\begin{aligned} &N_0(\xi,\eta,\zeta) = 1 - \xi^2 - \eta^2 - \zeta^2 \\ &N_1(\xi,\eta,\zeta) = -\frac{1}{2}\xi + \frac{1}{2}\xi^2 \qquad N_2(\xi,\eta,\zeta) = +\frac{1}{2}\xi + \frac{1}{2}\xi^2 \\ &N_3(\xi,\eta,\zeta) = -\frac{1}{2}\eta + \frac{1}{2}\eta^2 \qquad N_4(\xi,\eta,\zeta) = +\frac{1}{2}\eta + \frac{1}{2}\eta^2 \\ &N_5(\xi,\eta,\zeta) = -\frac{1}{2}\zeta + \frac{1}{2}\zeta^2 \qquad N_6(\xi,\eta,\zeta) = +\frac{1}{2}\zeta + \frac{1}{2}\zeta^2 \end{aligned}$$

The octahedron, or seven node star, as well as its 2D analogue, or ve node star, they both have been around in my work for quite some time:

```
http://hdebruijn.soo.dto.tudelft.nl/www/article/SUNA04.NET
http://hdebruijn.soo.dto.tudelft.nl/www/article/SUNA10.NET
http://hdebruijn.soo.dto.tudelft.nl/www/article/SUNA48.NET
http://hdebruijn.soo.dto.tudelft.nl/jaar2006/zevenpunt.jpg
```

Dual Core Element



Template for above picture copied without permission from:

```
http://en.wikipedia.org/wiki/File:Dual_Cube-Octahedron.svg
http://en.wikipedia.org/wiki/Octahedron
```

It's a well known fact that the octahedron is the dual polyhedron of the cube: see the wikipedia page for con rmation of this statement. In much the same way it can be said that the Finite Di erence like octahedron is the dual element of the Finite Element like (unit) hexahedron (cube). The two previous sections are related to each other in this way. And it is entirely in concordance with the good old Manifesto:

```
http://hdebruijn.soo.dto.tudelft.nl/www/programs/suna01.htm
http://hdebruijn.soo.dto.tudelft.nl/www/sunall.htm
```

It's a well established habit to use hexahedrons as the basic building blocks (bricks) in a three dimensional nite element mesh. Hexahedrons with arbitrary shape are derived from the unit cube by employing a so called *isoparametric transformation*. Let's repeat the shape functions of the hexahedron in the rst place:

$$\begin{aligned}
N_{h0}(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1-\zeta) \\
N_{h1}(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1-\zeta) \\
N_{h2}(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1-\zeta)
\end{aligned}$$

$$
\begin{aligned}
N_{h3}(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1-\zeta) \\
N_{h4}(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1+\zeta) \\
N_{h5}(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1-\eta)\frac{1}{2}(1+\zeta) \\
N_{h6}(\xi,\eta,\zeta) &= \frac{1}{2}(1-\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1+\zeta) \\
N_{h7}(\xi,\eta,\zeta) &= \frac{1}{2}(1+\xi)\frac{1}{2}(1+\eta)\frac{1}{2}(1+\zeta)
\end{aligned}
$$

$$f = N_{h0}f_{h0} + N_{h1}f_{h1} + N_{h2}f_{h2} + N_{h3}f_{h3} + N_{h4}f_{h4} + N_{h5}f_{h5} + N_{h6}f_{h6} + N_{h7}f_{h7}$$

Here the subscript h denotes that these (shape) functions belong to a **h**exahedron. An isoparametric transformation of the coordinates, i.e. $(\xi,\eta,\zeta) \rightarrow (x,y,z)$, is de ned in exactly the same way as with any other function at the element (hence the name "isoparametric" = with the same parameters):

$$
\begin{aligned}
x &= N_{h0}x_{h0} + N_{h1}x_{h1} + N_{h2}x_{h2} + N_{h3}x_{h3} + N_{h4}x_{h4} + N_{h5}x_{h5} + N_{h6}x_{h6} + N_{h7}x_{h7} \\
y &= N_{h0}y_{h0} + N_{h1}y_{h1} + N_{h2}y_{h2} + N_{h3}y_{h3} + N_{h4}y_{h4} + N_{h5}y_{h5} + N_{h6}y_{h6} + N_{h7}y_{h7} \\
z &= N_{h0}z_{h0} + N_{h1}z_{h1} + N_{h2}z_{h2} + N_{h3}z_{h3} + N_{h4}z_{h4} + N_{h5}z_{h5} + N_{h6}z_{h6} + N_{h7}z_{h7}
\end{aligned}
$$

This can be summarized in vector notation if we de ne $\vec{r} = (x,y,z)$:

$$\vec{r}(\xi,\eta,\zeta) = N_{h0}\vec{r}_{h0} + N_{h1}\vec{r}_{h1} + N_{h2}\vec{r}_{h2} + N_{h3}\vec{r}_{h3} N_{h4}\vec{r}_{h4} + N_{h5}\vec{r}_{h5} + N_{h6}\vec{r}_{h6} + N_{h7}\vec{r}_{h7}$$

Let's take a look now at the dual element of the general hexahedron, the parent of which is our unit octahedron. The places of the nodes of the unit octahedron inside the unit hexahedron are at:

$$
\begin{aligned}
(\xi_{o0},\eta_{o0},\zeta_{o0}) &= (0,0,0) \\
(\xi_{o1},\eta_{o1},\zeta_{o1}) &= (-1,0,0) \\
(\xi_{o2},\eta_{o2},\zeta_{o2}) &= (+1,0,0) \\
(\xi_{o3},\eta_{o3},\zeta_{o3}) &= (0,-1,0) \\
(\xi_{o4},\eta_{o4},\zeta_{o4}) &= (0,+1,0) \\
(\xi_{o5},\eta_{o5},\zeta_{o5}) &= (0,0,-1) \\
(\xi_{o6},\eta_{o6},\zeta_{o6}) &= (0,0,+1)
\end{aligned}
$$

Here the subscript o denotes that the unit coordinates belong to an **o**ctahedron. And, in case you didn't notice, this is a *staggered grid* with respect to the hexahedron's nodes. Then we nd (the trick is to remember that hexahedron numbering reads as coordinates in bits):

$$\vec{r}(0,0,0) = \frac{1}{8}(\vec{r}_{h0} + \vec{r}_{h1} + \vec{r}_{h2} + \vec{r}_{h3} + \vec{r}_{h4} + \vec{r}_{h5} + \vec{r}_{h6} + \vec{r}_{h7}) = \vec{r}_{o0}$$

$$\vec{r}(-1,0,0) = \frac{1}{4}(\vec{r}_{h0} + \vec{r}_{h2} + \vec{r}_{h4} + \vec{r}_{h6}) = \vec{r}_{o1}$$

$$\vec{r}(+1,0,0) = \frac{1}{4}(\vec{r}_{h1} + \vec{r}_{h3} + \vec{r}_{h5} + \vec{r}_{h7}) = \vec{r}_{o2}$$
$$\vec{r}(0,-1,0) = \frac{1}{4}(\vec{r}_{h0} + \vec{r}_{h1} + \vec{r}_{h4} + \vec{r}_{h5}) = \vec{r}_{o3}$$
$$\vec{r}(0,+1,0) = \frac{1}{4}(\vec{r}_{h2} + \vec{r}_{h3} + \vec{r}_{h6} + \vec{r}_{h7}) = \vec{r}_{o4}$$
$$\vec{r}(0,0,-1) = \frac{1}{4}(\vec{r}_{h0} + \vec{r}_{h1} + \vec{r}_{h2} + \vec{r}_{h3}) = \vec{r}_{o5}$$
$$\vec{r}(0,0,+1) = \frac{1}{4}(\vec{r}_{h4} + \vec{r}_{h5} + \vec{r}_{h6} + \vec{r}_{h7}) = \vec{r}_{o6}$$

We conclude herefrom that there exists a very simple relationship between the places of the nodes of the inner octahedron:

$$\vec{r}_{o0} = \frac{1}{2}(\vec{r}_{o1} + \vec{r}_{o2}) = \frac{1}{2}(\vec{r}_{o3} + \vec{r}_{o4}) = \frac{1}{2}(\vec{r}_{o5} + \vec{r}_{o6})$$

The question is if we can we infer the same relationship for any other function f of the normed coordinates. Of course we can, because we can derive in very much the same way as for the coordinates that:

$$f(0,0,0) = \frac{1}{8}(f_{h0} + f_{h1} + f_{h2} + f_{h3} + f_{h4} + f_{h5} + f_{h6} + f_{h7}) = f_{o0}$$
$$f(-1,0,0) = \frac{1}{4}(f_{h0} + f_{h2} + f_{h4} + f_{h6}) = f_{o1}$$
$$f(+1,0,0) = \frac{1}{4}(f_{h1} + f_{h3} + f_{h5} + f_{h7}) = f_{o2}$$
$$f(0,-1,0) = \frac{1}{4}(f_{h0} + f_{h1} + f_{h4} + f_{h5}) = f_{o3}$$
$$f(0,+1,0) = \frac{1}{4}(f_{h2} + f_{h3} + f_{h6} + f_{h7}) = f_{o4}$$
$$f(0,0,-1) = \frac{1}{4}(f_{h0} + f_{h1} + f_{h2} + f_{h3}) = f_{o5}$$
$$f(0,0,+1) = \frac{1}{4}(f_{h4} + f_{h5} + f_{h6} + f_{h7}) = f_{o6}$$

Consequently:

$$f_{o0} = \frac{1}{2}(f_{o1} + f_{o2}) = \frac{1}{2}(f_{o3} + f_{o4}) = \frac{1}{2}(f_{o5} + f_{o6})$$

But wait! Here is the F.D. representation of an arbitrary function interpolated at the octahedron:

$$\begin{aligned} f_o(\xi,\eta,\zeta) = f_{o0} &+ \frac{f_{o2} - f_{o1}}{2}\,\xi + (f_{o2} - 2f_{o0} + f_{o1})\,\frac{1}{2}\xi^2 \\ &+ \frac{f_{o4} - f_{o3}}{2}\,\eta + (f_{o4} - 2f_{o0} + f_{o3})\,\frac{1}{2}\eta^2 \\ &+ \frac{f_{o6} - f_{o5}}{2}\,\zeta + (f_{o6} - 2f_{o0} + f_{o5})\,\frac{1}{2}\zeta^2 \end{aligned}$$

So it turns out that, with the assumption of overall isoparametrics, all of the quadratic terms in the interpolation at the inner octahedron are cancelled. The resulting interpolation is *linear* :

$$f_o(\xi,\eta,\zeta) = f_{o0} + \frac{f_{o2}-f_{o1}}{2}\,\xi + \frac{f_{o4}-f_{o3}}{2}\,\eta + \frac{f_{o6}-f_{o5}}{2}\,\zeta$$

However, we can eliminate the odd node numbers, by substitution of:

$$f_{o0} = \frac{1}{2}(f_{o1}+f_{o2}) = \frac{1}{2}(f_{o3}+f_{o4}) = \frac{1}{2}(f_{o5}+f_{o6}) \quad \Longrightarrow$$

$$f_{o1} = 2f_{o0} - f_{o2} \quad \text{and} \quad f_{o3} = 2f_{o0} - f_{o4} \quad \text{and} \quad f_{o5} = 2f_{o0} - f_{o6}$$

Giving at last:

$$f_o(\xi,\eta,\zeta) = f_{o0} + (f_{o2}-f_{o0})\,\xi + (f_{o4}-f_{o0})\,\eta + (f_{o6}-f_{o0})\,\zeta$$

Thus the linear interpolation of an inner octahedron is exactly the same as the linear interpolation of one of the eight tetrahedrons composing that octahedron:

$$\begin{array}{cccc} (0,4,2,5) & (0,1,4,5) & (0,3,1,5) & (0,2,3,5) \\ (0,2,4,6) & (0,4,1,6) & (0,1,3,6) & (0,3,2,6) \end{array}$$

Doesn't matter which one is taken. Read the section "Linear Tetrahedron" in:

`http://hdebruijn.soo.dto.tudelft.nl/hdb_spul/belgisch.pdf`

Equations for Ideal Flow

For discretization of the equations for Ideal Flow, a dedicated version of the Least Squares Finite Element Method (L.S.FEM) will be employed. It has been argued for the two dimensional case that this so called nite element method is actually not so much a FEM but rather a Finite Volume Method. This is the reason why we will start with an integral (not a di erential) formulation of the equations governing ideal ow.
Consider a *Patch Test* element, consisting of just one hexahedron together with its inner octahedron. First we establish the number of unknowns, which is eight nodes times three velocity components, giving a total of 18. Then it shall be required that the total number of independent equations is equal to 18 as well. The rst equation is the one that de nes *incompressible* :

$$\iint (\vec{v}\cdot\vec{n})\,dA = 0$$

Here $\vec{v}$ is the ow velocity vector with components (u,v,w) , $\vec{n}$ is the normal on the surface A , and the integral is taken over this surface. In our case, the surface consists of the eight faces of the inner octahedron. Each of these faces is a triangle. Take one of the triangles, namely the one with vertices $(2,4,6)$.

Suppose that the interpolation of an arbitrary function f at this triangle is linear:

$$f(p,q) = f_2 + (f_4 - f_2)\,p + (f_6 - f_2)\,q$$

We know from the end of the preceding subsection:

$$f(\xi,\eta,\zeta) = f_0 + (f_2 - f_0)\,\xi + (f_4 - f_0)\,\eta + (f_6 - f_0)\,\zeta$$

Here we have dropped the subscript o , because the outer hexahedron is out of sight at the moment. When comparing this with the triangle formula for f:

$$f(p,q) = f_0 + (f_2 - f_0)(1 - p - q) + (f_4 - f_0)p + (f_6 - f_0)q$$

We see that, at the triangular surface:

$$p = \eta \quad \text{and} \quad q = \zeta \quad \text{and} \quad 1 - p - q = \xi \quad \Longrightarrow \quad \xi + \eta + \zeta = 1$$

So the in nitesimal area element is (where $\times$ denotes outer product):

$$\vec{n}\,dA = (\vec{r}_4 - \vec{r}_2) \times (\vec{r}_6 - \vec{r}_2)\,d\eta\,d\zeta$$

And the interpolation for the velocities is:

$$\vec{v} = \vec{v}_2 + (\vec{v}_4 - \vec{v}_2)\eta + (\vec{v}_6 - \vec{v}_2)\zeta$$

Now calculating the integral over that piece of the surface becomes a matter of routine (i.e. we have done this before):

$$\iint (\vec{v}\cdot\vec{n})\,dA = ((\vec{r}_4 - \vec{r}_2) \times (\vec{r}_6 - \vec{r}_2)) \cdot$$

$$\left[\vec{v}_2 \int_0^1 d\zeta \int_0^{1-\zeta} d\eta + (\vec{v}_4 - \vec{v}_2) \int_0^1 d\zeta \int_0^{1-\zeta} \eta\,d\eta + (\vec{v}_6 - \vec{v}_2) \int_0^1 \zeta d\zeta \int_0^{1-\zeta} d\eta\right]$$

$$= ((\vec{r}_4 - \vec{r}_2) \times (\vec{r}_6 - \vec{r}_2)) \cdot \left[\frac{1}{2}\vec{v}_2 + \frac{1}{6}(\vec{v}_4 - \vec{v}_2) + \frac{1}{6}(\vec{v}_6 - \vec{v}_2)\right] =$$

$$= \left[\frac{1}{2}(\vec{r}_4 - \vec{r}_2) \times (\vec{r}_6 - \vec{r}_2)\right] \cdot \frac{1}{3}\left[\vec{v}_2 + \vec{v}_4 + \vec{v}_6\right]$$

The above can also be derived from $\iint \xi^m \eta^n \, d\xi\, d\eta = m!n!/(m+n+2)!$ in:

`http://hdebruijn.soo.dto.tudelft.nl/jaar2004/simplex.pdf`

There are eight of these contributions, they must be summed up all together and the outcome must be zero. One (1) equation, that's all there is in this case. The second integral equation is the one that de nes *irrotational* :

$$\oint (\vec{v}\cdot d\vec{s}) = 0$$

It says that the *circulation* around each of the eight triangular faces is zero. With e.g. the above triangle, the interpolation at the edges is linear as well. To be precise:

$$\begin{array}{llllll} \text{at} & (\vec{r}_4 - \vec{r}_2) & : & \xi + \eta = 1 & \text{and} & \zeta = 0 \\ \text{at} & (\vec{r}_6 - \vec{r}_4) & : & \eta + \zeta = 1 & \text{and} & \xi = 0 \\ \text{at} & (\vec{r}_2 - \vec{r}_6) & : & \xi + \zeta = 1 & \text{and} & \eta = 0 \end{array}$$

So the contour integral is discretized as:

$$((\vec{r}_4 - \vec{r}_2) \cdot \frac{1}{2}(\vec{v}_4 + \vec{v}_2)) + ((\vec{r}_6 - \vec{r}_4) \cdot \frac{1}{2}(\vec{v}_6 + \vec{v}_4)) + ((\vec{r}_2 - \vec{r}_6) \cdot \frac{1}{2}(\vec{v}_2 + \vec{v}_6)) = 0$$

And in the same way for all eight faces of the inner octahedron. So it seems, at rst sight, that we have eight equations here. But only at rst sight. Taking a further look at the problem reveals that, due to common edges of the faces, there are only four (4) independent equations. You can see this by drawing the arrows of four (non adjacent) circulation patterns and note that all edges then already *have* a circulation.
The velocity components are interpolated by isoparametric (ξ, η, ζ) coordinates and therefore are subject to the same relations that have been established for any function at the inner octahedron. This is what we have found:

$$f_{o0} = \frac{1}{2}(f_{o1} + f_{o2}) = \frac{1}{2}(f_{o3} + f_{o4}) = \frac{1}{2}(f_{o5} + f_{o6})$$

Consequently:

$$\vec{v}_0 = \frac{1}{2}(\vec{v}_1 + \vec{v}_2) = \frac{1}{2}(\vec{v}_3 + \vec{v}_4) = \frac{1}{2}(\vec{v}_5 + \vec{v}_6)$$

Here we have dropped the subscript o again, because the outer hexahedron is out of sight at the moment. Upon eliminating $\vec{v}_0$ we have:

$$\begin{array}{lcl} \frac{1}{2}(\vec{v}_1 + \vec{v}_2) & = & \frac{1}{2}(\vec{v}_3 + \vec{v}_4) \\ \frac{1}{2}(\vec{v}_1 + \vec{v}_2) & = & \frac{1}{2}(\vec{v}_5 + \vec{v}_6) \\ \frac{1}{2}(\vec{v}_3 + \vec{v}_4) & = & \frac{1}{2}(\vec{v}_5 + \vec{v}_6) \end{array}$$

It is clear that, for example, the third of these equations is dependent on the other two. So e ectively there are two vector equations times three components giving six (6) independent equations.
At last we have Boundary Conditions. These consist of impermeable walls at the nodes $(3, 4, 5, 6)$, a prescribed velocity (three components) at the "inlet" node (1) and nothing at the "outlet" node (2). Leading to 4 + 3 = seven (7) independent equations.
So what we have in total is 1 + 4 + 6 + 7 = 18 , which is exactly the number of unknowns, hence the required number of independent Finite Volume equations.

Delphi Pascal Source Code

```
program stroming;
{
  Patch Test for Linear Octahedron with
  Incompressible and Irrotational Flow
  (Ideal Flow) in three dimensions: 3-D
}
Uses Numeriek; { Numerical Toolbox @ website }

type
  vektor = record
    x,y,z : double;
  end;
  xyz = (x,y,z); { Enumeration }

const
{ All 6 vertices of the octahedron }
  P : array[0..6,0..2] of double =
  (( 0, 0, 0) { 0 }
  ,(-1, 0, 0) { 1 }
  ,(+1, 0, 0) { 2 }
  ,( 0,-1, 0) { 3 }
  ,( 0,+1, 0) { 4 }
  ,( 0, 0,-1) { 5 }
  ,( 0, 0,+1) { 6 } );

{ All eight faces of the octahedron }
  nr : array[0..7,0..2] of integer =
  ((2,4,6),(4,1,6),(1,3,6),(3,2,6)
  ,(4,2,5),(1,4,5),(3,1,5),(2,3,5));

{ Smart numbering does the job, a great deal! }

function no(node : integer; speed : xyz) : integer;
{
  Numbering within FEM
}
begin
  no := (node-1)*3 + Ord(speed) + 1;
end;

var
  FEM : Symmetric_Band_Positive_Incore;

function pijl(one,two : integer) : vektor;
```

```
{
  Edge of Octahedron
}
var
  v : vektor;
begin
  v.x := P[two,0] - P[one,0];
  v.y := P[two,1] - P[one,1];
  v.z := P[two,2] - P[one,2];
  pijl := v;
end;

function uit(a,b : vektor) : vektor;
{
  Outer Product
}
var
  u : vektor;
begin
  u.x := a.y*b.z - a.z*b.y;
  u.y := a.z*b.x - a.x*b.z;
  u.z := a.x*b.y - a.y*b.x;
  uit := u;
end;

procedure Incompressible;
{
  Incompressible Flow
}
var
  i,j,m : integer;
  bij : vektor;
begin
  FEM.Schema_nul(18);
  for i := 0 to 7 do
  begin
  { Triangular Face contribution }
    bij := uit(pijl(nr[i,0],nr[i,1])
              ,pijl(nr[i,0],nr[i,2]));
  { Do the bookkeeping }
    for j := 0 to 2 do
    begin
      m := no(nr[i,j],x); FEM.A[m] := FEM.A[m] + bij.x;
      m := no(nr[i,j],y); FEM.A[m] := FEM.A[m] + bij.y;
      m := no(nr[i,j],z); FEM.A[m] := FEM.A[m] + bij.z;
    end;
```

```
  end;
  FEM.Element_nul(18);
  FEM.Kleinste_Kwadraten(18,1);
  FEM.Intellen(18);
{ Count = 1 }
end;

procedure Irrotational;
{
  Irrotational Flow
}
var
  i,j,k,L,m : integer;
  bij : vektor;
begin
  for i := 0 to 7 do
  begin
    FEM.Schema_nul(18);
    for k := 0 to 2 do
    begin
    { Triangle Edges contribution }
      L := (k+1) mod 3;
      bij := pijl(nr[i,k],nr[i,L]);
    { Do the bookkeeping }
      for j := 0 to 1 do
      begin
        L := (k+j) mod 3;
        m := no(nr[i,L],x); FEM.A[m] := FEM.A[m] + bij.x;
        m := no(nr[i,L],y); FEM.A[m] := FEM.A[m] + bij.y;
        m := no(nr[i,L],z); FEM.A[m] := FEM.A[m] + bij.z;
      end;
    end;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);
  { Count = 8 / 2 = 4 -> 5 }
  end;
end;

procedure Boundaries;
{
  Boundary Conditions
}
var
  m : integer;
  yz : xyz;
```

```
begin
{ Prescribed velocity (1,0,0) at (1) }
  FEM.Schema_nul(18);
  FEM.A[no(1,x)] := 1; FEM.rhs := 1;
  FEM.Element_nul(18);
  FEM.Kleinste_Kwadraten(18,1);
  FEM.Intellen(18);
  for yz := y to z do
  begin
    FEM.Schema_nul(18);
    FEM.A[no(1,yz)] := 1;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);
  end;
{ Walls in Y direction }
  for m := 3 to 4 do
  begin
    FEM.Schema_nul(18);
    FEM.A[no(m,y)] := 1;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);
  end;
{ Walls in Z direction }
  for m := 5 to 6 do
  begin
    FEM.Schema_nul(18);
    FEM.A[no(m,z)] := 1;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);
  end;
{ Count = 3 + 4 = 7 -> 12 }
end;

procedure Isoparametrics;
{
  Isoparametrics of
  Linear Octahedron
}
var
  v : xyz;
begin
  for v := x to z do
  begin
```

```
    FEM.Schema_nul(18);
    FEM.A[no(1,v)] := +1;
    FEM.A[no(2,v)] := +1;
    FEM.A[no(3,v)] := -1;
    FEM.A[no(4,v)] := -1;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);

    FEM.Schema_nul(18);
    FEM.A[no(1,v)] := +1;
    FEM.A[no(2,v)] := +1;
    FEM.A[no(5,v)] := -1;
    FEM.A[no(6,v)] := -1;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);

    FEM.Schema_nul(18);
    FEM.A[no(2,v)] := +1;
    FEM.A[no(3,v)] := +1;
    FEM.A[no(5,v)] := -1;
    FEM.A[no(6,v)] := -1;
    FEM.Element_nul(18);
    FEM.Kleinste_Kwadraten(18,1);
    FEM.Intellen(18);
  end;
{ Count = (3-1) * 3 = 6 -> 18 }
end;

procedure doen;
{
  Patch Test
  Do the Job
}
var
  k : integer;
begin
{ Initialize }
  FEM.nn := 18;
  FEM.nb1 := 18;
  FEM.Globaal_nul;
  for k := 1 to 18 do
    FEM.nr[k] := k;
{ All Equations }
  Incompressible;
```

```
  Irrotational;
  Isoparametrics;
  Boundaries;
{ Solution }
  FEM.Oplossen('flowin3D.txt');
end;

procedure test;
{
  Just a test
}
var
  k : integer;
begin
  for k := 1 to 6 do
  begin
    Writeln(no(k,x));
    Writeln(no(k,y));
    Writeln(no(k,z));
  end;
end;

begin
  doen;
end.
```

Results of Patch Test

The results are ordered as

u_1	v_1	w_1	u_2	v_2	w_2	u_3	v_3	w_3	u_4	v_4	w_4	u_5	v_5	w_5	u_6	v_6	w_6
1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18

Where (u_k, v_k, w_k) are ow velocity components at node (k) of the octahedron.

```
 1  1.00000000000000E+0000
 2  4.56485884640815E-0017
 3 -7.53265120626704E-0018
 4  1.00000000000000E+0000
 5  6.12595591043916E-0017
 6 -5.30319498511005E-0018
 7  1.00000000000000E+0000
 8  1.39043050177858E-0016
 9 -1.93468035350552E-0017
10  1.00000000000000E+0000
11 -1.01886958795359E-0016
12  1.85589355162947E-0017
```

```
   13  1.00000000000000E+0000
   14  5.28281767444885E-0017
   15  5.23664259526328E-0017
   16  1.00000000000000E+0000
   17  9.86155590864014E-0017
   18 -6.37942359686740E-0017
```

It is seen that the prescribed velocity $(u_1, v_1, w_1) = (1, 0, 0)$ at the entrance is copied everywhere in the flow field, as it should be.

Disclaimers

Anything free comes without referee :-(
My English may be better than your Dutch.